#!/usr/bin/env python
from misc.utility.scons_hints import *

Import("env")

import os

import editor_icons_builders

# Editor's own icons
icon_sources = Glob("*.svg")

# Module icons
for path in env.module_icons_paths:
    if not os.path.isabs(path):
        icon_sources += Glob(f"#{path}/*.svg")  # Built-in.
    else:
        icon_sources += Glob(f"{path}/*.svg")  # Custom.

env.CommandNoCache(
    "#editor/themes/editor_icons.gen.h",
    icon_sources,
    env.Run(editor_icons_builders.make_editor_icons_action),
)
